

Упражнение 2

Рекурсивни функции: рекурсия и итерация.

Функциите могат да генерират **рекурсивен** или **итеративен** процес на изчисление.

При рекурсивния имаме отложени изчисления до връщане на резултат от рекурсивното прилагане на функцията. При итеративния няма отложени изчисления, а на всяка стъпка се получава частичен резултат, който се доближава до търсения, получаван на последната стъпка. И в двата случая функциите са рекурсивни.

При реализиране на итерация чрез рекурсивна функция се слагат допълнителни параметри на функцията, които при всяко извикване актуализират стойността си.

2.1. Дефинирайте функция, която пресмята произведението: $\prod_{i=1}^k \frac{2i+1}{(2i-3)^2}$.

Рекурсивно решение (линейна рекурсия):

```
(define (pr1-rec k)
  (if (= k 1) 3
      (* (pr1-rec (- k 1)) (/ (+ (* 2 k) 1) (expt (- (* 2 k) 3) 2)))))
```

Итеративно решение:

```
(define (pr1 k)
  (define (iter-pr1 i p)
    (if (> i k) p
        (iter-pr1 (+ i 1) (* p (/ (+ (* 2 i) 1) (expt (- (* 2 i) 3) 2))))))
  (iter-pr1 1 1))
```

2.2. Дефинирайте функция, която пресмята произведението: $\prod_{j=0}^n \frac{(x+j)^3}{3}$.

Рекурсивно решение:

```
(define (pr2-rec n x)
  (if (= n 0) (/ (expt x 3) 3)
      (* (pr2-rec (- n 1) x) (/ (expt (+ x n) 3) 3))))
```

Итеративно решение:

```
(define (pr2 n x)
  (define (iter-pr2 j p)
    (if (> j n) p (iter-pr2 (+ j 1) (* p (/ (expt (+ x j) 3) 3)))))
  (iter-pr2 0 1))
```

2.3. Дефинирайте функция, която пресмята сумата: $\sum_{j=0}^k \frac{(j+x)^2-2}{4}$.

2.4. Дефинирайте функция, която по зададено n пресмята n-тия член на редицата: $a_1=6; a_2=-4; a_n=3.a_{n-1}-2.a_{n-2}.n.(n-1), n>2$.

Рекурсивно решение (дървовидна рекурсия):

```
(define (red-rec n)
  (cond ((= n 1) 6)
        ((= n 2) -4)
        ((> n 2) (- (* 3 (red-rec (- n 1))) (* 2 (red-rec (- n 2)) n (- n 1)))))
  ))
```

Итеративно решение:

```
(define (red n)
  (define (red-iter k b c)
    (if (> k n) c
        (red-iter (+ k 1) c (- (* 3 c) (* 2 b k (- k 1)))))
  )
  (if (= n 1) 6 (red-iter 3 6 -4))
)
```

2.5. Дефинирайте функция за пресмятане на n-ти елемент на редицата на Фибоначи.

Итеративно решение:

```
(define (fibon n)
  (define (iter k x y)
    (if (= k n) y (iter (+ k 1) y (+ x y))))
  (iter 0 0 1))
```

2.6. Дефинирайте функция, която намира най-малкото число от редицата на Фибоначи, което е по-голямо от зададено n.

```
(define (fibN n)
  (define (iter x y)
    (if (> y n) y (iter y (+ x y))))
  (iter 1 1))
```

2.7. Дефинирайте функция, която по зададени реално x и цяло n намира стойността за x на n-ия полином от редицата:

$$P_0(x)=1, P_1(x)=x, P_n(x)=((2.n-1).x.P_{n-1}(x)-(n-1).P_{n-2}(x))/n, n>1.$$

Рекурсивно решение:

```
(define (p-rec x n)
  (cond ((= n 0) 1)
        ((= n 1) x)
        ((> n 1) (/ (- (* (- (* 2 n) 1) x (p-rec x (- n 1)))
                        (* (- n 1) (p-rec x (- n 2))))
                     n)))
  ))
```

Итеративно решение:

```
(define (p x n)
  (define (cycle k y z)
    (if (> k n) z
        (cycle (+ k 1) z (/ (- (* (- (* 2 k) 1) x z) (* (- k 1) y)) k)))
  )
  (if (> n 0) (cycle 2 1 x) 1)
)
```

2.8. Дефинирайте функция, която намира i-ия елемент на редицата:

$$c_1=2, c_i = \frac{c_{i-1} + 3i - 8}{i - 1}, i > 1.$$

2.9. Дефинирайте функция, която намира k-ия елемент на редицата:

$$a_1=2, a_2=4, a_k = \frac{0,5.k.a_{k-1} - 3,5.a_{k-2}}{2.k-1}, k > 2.$$

2.10. Дефинирайте функция с параметри n и x, която генерира линейно итеративен процес за намиране на стойността на полинома:

$$P(x, n) = n.x^n + (n-1).x^{n-1} + \dots + 2.x^2 + x.$$

2.11. Дефинирайте функция, която намира десетичното число, което се получава при записване на десетичните цифри на естествено число n в обратен ред (водещите нули се игнорират).

```
(define (rev n)
  (define (iter x s)
    (if (= x 0) s (iter (quotient x 10) (+ (remainder x 10) (* s 10)))) )
  (iter n 0)
)
```

2.12. Дефинирайте функция, която при дадено n проверява дали в редицата числа $i^4 + 3.i.n^2 + n^4$, $i=1, \dots, n$ има такова, което се дели на x.